
                           9S10A

                   TEST FOR WALKING SWITCH
                   CORES IN 738 CORE ON 709

A. PURPOSE

   This test is to be used on a 709 system with 738 core storage, It attempts to
   aggravate 5998’s which continue to have small plate currents when the grid
   input is down. This small plate current may "walk" a half select switch core.

B. AREA OF MACHINE REQUIRED
   1. Units Card Reader
          Main Frame with 738

   2. Storage Locations
          Program 00000 to 00027
          Test area 00030 to 77777

C. PROGRAM CONTROL
   1. Deck - one card 9S10A00
               Self loading
   2. No Sense Switches
   3. No Sense Lite

D. ERROR STOPS
   00007 - Core position about to be tested is not clear. Switch Core "walked"!
   Address is in accumulator

   00016 - Core position just tested can not now be cleared. Possible bad switch
   core. Address is in accumulator.

E. PRINT-OUTS
   None

F. COMMENTS
   The STORE and subsequent EXECUTE of the CAQ with the decrement indicated
   causes 76 consecutive "E" cycles to the same address. This will half select other
   switch cores 76 times and unstable 5998’s may then "walk" a switch core.
   It is possible to marginal test the -220V MSD-DPD bias at -216V#l to aggravate
   the 5998’s. This control is located below the MSD Reference Voltage Adjust in
   CSF-1. The MSD-DPD bias may be lowered to -222V as the spec states it to be
   -220V+/-2V. This procedure increases the reliability of the 5998 tube bias.

       REM

       REM

       FUL

START  TXL 3,0,23       IOST FOR READ CARD

ONE    TCOA *           HOLD TILL IN

       STD *-1          MAKE 00001 = 1

       CAL TEMP-1       SECURE TEST WORD

       SLW TEMP         TEMP STORAGE

XRA    AXT START-24,1   LOAD INDEX

       ZET 0,1          CHECK TEST LOCATION

       HPR 0,1          FOR ZERO - NO!

       STO 0,1          YES - STORE TEST LOC.

       ADD ONE          INCREASE ADDRESS

       SLW TEMP         SAVE AT TEMP

       XEC 0,1          EXECUTE CAQ WITH 76 "E" CYCLES

       STZ 0,1          CLEAR

       ZET 0,1          TEST ZERO

       HPR 0,1          NO - CANNOT RESTORE

       CAL TEMP         OKAY

       TIX XRA+1,1,1    NEXT TEST LOCATION

       TRA START+3      REPEAT TEST

       CAQ *+6,0,62     TEST WORD - 76 E CYCLES

TEMP   HTR 0

       END

source source

           00000 -3 00027 0 00003  START  TXL 3,0,23       IOST FOR READ CARD
           00001  0060 00 0 00001  ONE    TCOA *           HOLD TILL IN
           00002  0622 00 0 00001         STD *-1          MAKE 00001 = 1
           00003 -0500 00 0 00022         CAL TEMP-1       SECURE TEST WORD
           00004  0602 00 0 00023         SLW TEMP         TEMP STORAGE
           00005  0774 00 1 77750  XRA    AXT START-24,1   LOAD INDEX
           00006  0520 00 1 00000         ZET 0,1          CHECK TEST LOCATION
           00007  0420 00 1 00000         HPR 0,1          FOR ZERO - NO!
           00010  0601 00 1 00000         STO 0,1          YES - STORE TEST LOC.
           00011  0400 00 0 00001         ADD ONE          INCREASE ADDRESS
           00012  0602 00 0 00023         SLW TEMP         SAVE AT TEMP
           00013  0522 00 1 00000         XEC 0,1          EXECUTE CAQ WITH 76 }E} CYCLES
           00014  0600 00 1 00000         STZ 0,1          CLEAR
           00015  0520 00 1 00000         ZET 0,1          TEST ZERO
           00016  0420 00 1 00000         HPR 0,1          NO - CANNOT RESTORE
           00017 -0500 00 0 00023         CAL TEMP         OKAY
           00020  2 00001 1 00006         TIX XRA+1,1,1    NEXT TEST LOCATION
           00021  0020 00 0 00003         TRA START+3      REPEAT TEST
           00022 -0114 76 0 00030         CAQ *+6,0,62     TEST WORD - 76 E CYCLES
           00023  0000 00 0 00000  TEMP   HTR 0
                            00000         END

